

教科書 13 章

13. 1. 2 設計例 1 と設計例 2 のプログラムを Scilab で作成した。これらを以下に示す。実行には関数プログラム `bode2.sci` が必要である。

佐伯正美 広島大学
2014年5月10日

1) 設計例 1

```
// 13. 1 節 設計例 1
// sec13_1_example1.sce M. Saeki, May 2014
// 古典制御による設計
clear
exec('bode2.sci',-1);
//
s=poly(0,'s');
sysP=syslin('c',1/(s^3+s^2));
N=300;
w=logspace(-2,1,N);
M=200;
t=linspace(0,60,M);

// KP の設計
scf(1);
bode2(sysP,w);
// 位相進みの設計
wmax=0.3;
pmax=60*%pi/180;
T1=sqrt((1-sin(pmax))/(1+sin(pmax)))/wmax;
T2=sqrt((1+sin(pmax))/(1-sin(pmax)))/wmax;
K1=(1+s*T2)/(1+s*T1)//位相進み要素
sysP2=sysP*K1
scf(1);
bode2(sysP2,w);
// ゲイン調整
sysP22=sysP*(10^(-32/20))*K1
scf(2);
```

```
bode2(sysP22,w)
```

```
// フィードバック系のステップ応答
```

```
sysS22=1/(1+sysP22);
```

```
sysT22=1-sysS22
```

```
scf(3);
```

```
y=csim('step',t,sysT22);
```

```
plot(t,y)
```

```
xtitle('step response','time[s]','y')
```

```
xgrid();
```

設計例 2)

```
// 13. 1節 設計例 2
```

```
// sec13_1_example2.sce M. Saeki, May 2014
```

```
// 古典制御による設計
```

```
clear
```

```
exec('bode2.sci',-1);
```

```
//
```

```
s=poly(0,'s');
```

```
sysP=syslin('c',1/((s+2)*(s+1)));
```

```
N=300;
```

```
w=logspace(-2,1,N);
```

```
M=200;
```

```
t=linspace(0,20,M);
```

```
// 積分制御 KI/s の設計
```

```
sysL1=sysP/s;
```

```
scf(1);
```

```
bode2(sysL1,w);
```

```
// 位相進みの設計
```

```
wmax=2;
```

```
pmax=60*%pi/180;
```

```
T1=sqrt((1-sin(pmax))/(1+sin(pmax)))/wmax;
```

```
T2=sqrt((1+sin(pmax))/(1-sin(pmax)))/wmax;
```

```
K1=(1+s*T2)/(1+s*T1)//位相進み要素
```

```

sysL2=sysL1*K1
scf(1);
bode2(sysL2,w);
sysL22=sysL2*(10^(10/20));// ゲイン調整
scf(2);
bode2(sysL22,w)
// フィードバック系のステップ応答
sysS22=1/(1+sysL22);
sysT22=1-sysS22
scf(3);
y=csim('step',t,sysT22);
plot(t,y)
xlabel('step response','time[s]','y')
xgrid();

// 位相遅れの設計
T4=10
T3=3*T4
K2=(1+s*T4)*T3/((1+s*T3)*T4)
sysL3=sysL22*K2
scf(2);
bode2(sysL3,w)
// フィードバック系のステップ応答
sysS3=1/(1+sysL3);
sysT3=1-sysS3
scf(3);
y=csim('step',t,sysT3);
plot(t,y)
xlabel('step response','time[s]','y')
xgrid();
// フィードバック系のランプ応答の比較
sysT3ramp=sysT3/s;
scf(4);
y3=csim('step',t,sysT3ramp);
sysT22ramp=sysT22/s;
y22=csim('step',t,sysT22ramp);

```

```
plot(t,y3,t,y22)
xtitle('step response','time[s]','y')
xgrid();
```